

SCIENCE & VIE

MENSUEL

N° 825 JUIN 1986

**RADIOACTIVITÉ:
MESUREZ
VOUS-MÊME**



**COMPACT-DISC : 11 APPAREILS
AU BANC D'ESSAI**

● L'astrologie
vérifiée

● Eve
était
noire

16 F
N° 825
SUISSE 5 FS
CANADA \$ 2,95
BELGIQUE 110 FB
ESPAGNE 450 Ptas
MAROC 17 Dh
TUNISIE 1,6 DT

ISSN 0036 8369

Dessinez vos sons

INFORMATIQUE AMUSANTE

Les possibilités de création de sons sont nombreuses sur l'Amstrad, mais la détermination des paramètres souvent fastidieuse. Nous vous proposons donc de réaliser un petit programme d'assistance pour leur calcul. Nous ne nous attacherons ici qu'à la commande d'enveloppe, c'est-à-dire à l'évolution du volume du son dans le temps. Il suffira de dessiner la courbe (en réalité, une ligne brisée) de l'amplitude sonore en fonction du temps sur l'écran pour que le programme prenne en charge le calcul de l'ensemble des paramètres lui correspondant.

L'ordre basic utilisé pour faire assimiler ces données par l'ordinateur sera ENV. Bien que celui-ci permette certaines libertés, tous les cas de figure ne sont pas possibles. Au fur et à mesure du tracé de votre enveloppe, le programme donnera donc sur l'écran le gabarit d'enveloppe se rapprochant le plus de celui demandé.


Le résultat sonore obtenu sera fourni en fin de programme. Ce dernier contrôle permettra de vérifier si le résultat obtenu corres-

pond bien à celui souhaité. Passons à présent à l'écriture du programme. En premier lieu, nous définirons le caractère graphique qui sera utilisé pour afficher le quadrillage de fond sur l'écran durant le tracé de l'enveloppe — lignes 40 et 50 — puis nous effacerons l'écran et la machine sera positionnée en mode

2, de manière à obtenir une surface d'affichage suffisante pour l'inscription des divers commentaires — ligne 60. Vient ensuite l'initialisation des variables; elle occupe les lignes 100 à 190. AP, AV et AD correspondant respectivement au nombre de pas, au volume de chaque pas et à sa durée pour l'attaque. De même MP, MV et MD concerneront le maintien et FP, FV et FD la décroissance. UP sera utilisée pour interdire le déplacement vertical du curseur durant le tracé de la décroissance.

La grille est alors tracée sur l'écran. La boucle double FOR... NEXT des lignes 230 à 270 encadrant l'instruction PRINT assure cette opération. La durée approximative du son désiré sera demandée par l'ordre INPUT de la ligne 310 et la variable DD prendra une valeur que nous utiliserons ultérieurement en tant qu'échelle de temps. L'ensemble des paramètres ainsi déterminé sera résumé sur le bord de l'écran par les lignes 350 à 420. Le tracé de l'enveloppe peut donc commencer.

Nous commencerons, c'est lo-



pond bien à celui souhaité. Passons à présent à l'écriture du programme.

En premier lieu, nous définirons le caractère graphique qui sera utilisé pour afficher le quadrillage de fond sur l'écran durant le tracé de l'enveloppe — lignes 40 et 50 — puis nous effacerons l'écran et la machine sera positionnée en mode

gique, par l'attaque. Le programme se rendra à la sous-routine 1360. Celle-ci commande les déplacements du curseur sur la grille en fonction des flèches de mouvement tapées. Deux nouvelles variables, HC et VC, seront utilisées. HC correspond à la position horizontale du curseur et VC à la verticale. HC sera donc incrémenté,

ou décrémente, en fonction de la frappe "flèche à droite" ou "flèche à gauche". La même opération sera pratiquée sur VC mais en fonction, cette fois-ci, de "flèche en haut" ou "flèche en bas". Nous trouverons ces deux opérations en lignes 1380 et 1390, on notera — en ligne 1390 — la présence de la variable UP inhibant l'effet de la ligne lors du tracé de la décroissance.

De 1400 à 1430 quatre tests seront effectués sur HC et VC afin de vérifier si celles-ci sont bien compatibles avec les possibilités d'affichage du curseur sur notre grille. Enfin si "ENTER" est tapé, touche que nous utiliserons pour valider les coordonnées choisies, le programme sortira de la sous-routine, en cas contraire il rebouclera sur cette dernière afin de permettre d'autres déplacements du curseur.

Le programme reviendra donc sur la détermination de l'attaque. Deux cas sont examinés en fonction des valeurs prises par HC et VC. Ils permettront de déterminer si la pente choisie pour l'attaque est supérieure à 45 degrés ou non. En fonction de cela les valeurs des variables AP, AV et AD seront calculées. Le tracé sera alors effectué par les lignes 550 à 590. Le résultat présenté ne correspond pas obligatoirement exactement à celui choisi : il est l'image du cas possible se rapprochant le plus de vos désirs. Les paramètres correspondants à l'attaque représentée seront résumés par la ligne 630, puis le programme passera à la détermination du maintien.

HCP et VCP seront employées pour mémoriser la position du curseur correspondant à la fin de l'attaque puis la sous-routine 1360 sera de nouveau utilisée pour le déplacer. La durée du maintien ne pouvant être négative, un test sur la nouvelle valeur prise par HC, après déplacement du curseur, sera effectué. Toute proposition impossible sera donc refusée par le programme. Ensuite, là encore en fonction des valeurs prises par HC et VC, divers cas seront étudiés, à savoir si la pente d'évolution choisie est nulle, positive ou négative. Ces trois tests sont effectués des lignes 750 à 770. En fonction des résultats le tracé des points pourra être entrepris selon un procédé proche de celui utilisé pour l'attaque. Les résultats obtenus seront résumés par la ligne

```

10 REM *****
20 REM *   DEF. DU CAR. DE LA GRILLE   *
30 REM *****
40 SYMBOL AFTER 254
50 SYMBOL 255,255,129,129,129,129,129,129,255
60 CLS:MODE 2:CLEAR
70 REM *****
80 REM *   INITIALISATION DES VARIABLES   *
90 REM *****
100 LET AP=0
110 LET AV=0
120 LET AD=0
130 LET MP=0
140 LET MV=0
150 LET MD=0
160 LET FP=0
170 LET FV=0
180 LET FD=0
190 LET UP=1
200 REM *****
210 REM *   AFFICHAGE DE LA GRILLE   *
220 REM *****
230 FOR V=1 TO 15
240 FOR H=1 TO 50
250 LOCATE H,V:PRINT CHR$(255)
260 NEXT H
270 NEXT V
280 REM *****
290 REM *   CHOIX DE L'AXE DES TEMPS   *
300 REM *****
310 LOCATE 1,20:INPUT "DUREE TOTALE DE LA NOTE EN SECONDES ";D
320 LET DD=2*D
330 LOCATE 1,20:PRINT "
"
340 IF DD<1 THEN GOTO 310
350 LOCATE 55,1:PRINT "DUREE PAR DIVISION:"
360 LOCATE 55,2:PRINT DD;"CENTIEMES DE SECONDE"
370 LOCATE 55,5:PRINT "UTILISER LES FLECHES "
380 LOCATE 55,6:PRINT "POUR DEPLACER LE CURSEUR. "
390 LOCATE 55,7:PRINT "VALIDEZ VOS POINTS"
400 LOCATE 55,8:PRINT "EN TAPANT ENTER. "
410 LOCATE 55,12:PRINT "DUREE TOTALE DE"
420 LOCATE 55,13:PRINT "L'ECRAN:";D;"SECONDES. "
430 REM *****
440 REM *   DETERMINATION DE L'ATTAQUE   *
450 REM *****
460 LET HC=1:LET VC=1
470 GOSUB 1360
480 IF VC>=HC THEN LET AP=HC:LET AV=INT(VC/HC):LET AD=DD
490 IF VC<HC THEN LET AP=VC:LET AV=1:LET AD=DD*INT(HC/VC)
500 LOCATE HC,16-VC:PRINT CHR$(255)
510 LOCATE (AP*AD)/DD,16-(AV*AP):PRINT CHR$(233)
520 REM *****
530 REM *   TRACE DES POINTS   *
540 REM *****
550 FOR I=1 TO AP*(AD/DD)
560 FOR J=1 TO AV
570 LOCATE I,16-(J+INT(AV*(DD/AD)*(I-1))):PRINT CHR$(233)
580 NEXT J
590 NEXT I
600 REM *****
610 REM *   AFFICHAGE DU RESULTAT   *
620 REM *****
630 LOCATE 1,17:PRINT"- ATTAQUE:";AP;"PAS DE";AV;"ET D'UNE DUREE DE";AD;"CENTIEMES DE SECONDE CHACUNS. "
640 REM *****
650 REM *   DETERMINATION DU MAINTIEN   *
660 REM *****
670 LET HC=(AP*AD)/DD:LET VC=(AV*AP)
680 LET HCP=HC:LET VCP=VC
690 GOSUB 1360
700 REM *****
710 REM *   TRACE DES POINTS   *
720 REM *****
730 IF HC<=HCP THEN GOTO 690
740 IF (HC-HCP)<=ABS(VCP-VC) THEN GOTO 840
750 IF VC=VCP THEN LET MV=0:LET MP=1:LET MD=(HC-HCP)*DD
760 IF VC>VCP THEN LET MV=1:LET MP=ABS(VC-VCP):LET MD=DD*INT(ABS(HC-HCP)/(VC-VCP))
770 IF VC<VCP THEN LET MV=-1:LET MP=ABS(VC-VCP):LET MD=DD*INT(ABS(HC-HCP)/(VC-VCP))

```

```

780 LOCATE HC,16-VC:PRINT CHR$(255)
790 LOCATE (AP*(AD/DD))+((MP*MD)/DD),16-VCP-(MV*MP):PRINT CHR$(
233)
800 FOR I=0 TO (MP*(MD/DD))
810 LOCATE I+(AP*(AD/DD)),16-(AV*AP)-INT(I*(DD/MD)*MV):PRINT CH
R$(233)
820 NEXT I
830 GOTO 910
840 LET MP=HC-HCP:LET MV=INT(0.5+((VC-VCP)/(HC-HCP))):LET MD=DD
850 FOR I=0 TO MP STEP ABS(1/(MV))
860 LOCATE I+(AP*(AD/DD)),16-(AV*AP)-INT(I*(DD/MD)*MV):PRINT CH
R$(233)
870 NEXT I
880 REM *****
890 REM * AFFICHAGE DES RESULTATS *
900 REM *****
910 LOCATE 1,18:PRINT"- MAINTIEN:";MP;"PAS DE";MV;"ET D'UNE DUR
EE DE";MD;"CENTIEMES DE SECONDE CHACUNS."
920 REM *****
930 REM * DETERMINATION DE LA DECROISSANCE *
940 REM *****
950 LET UP=0:LET HCP=INT((AP*AD/DD)+(MP*MD/DD)):LET VCP=VC:LET
VC=1
960 GOSUB 1360
970 IF HC<=HCP THEN GOTO 960
980 REM *****
990 REM * TRACE DES POINTS *
1000 REM *****
1010 IF (HC-HCP)<VCP THEN LET FV=-INT(VCP/(HC-HCP)):LET FD=DD
1020 IF (HC-HCP)>=VCP THEN LET FV=-1:LET FD=DD*INT((HC-HCP)/VCP
)
1030 LET FP=INT(((AP*AV)+(MP*MV))/-FV)
1040 LOCATE HC,15:PRINT CHR$(255)
1050 FOR I=1 TO INT(FP*(FD/DD))
1060 FOR J=1 TO -FV
1070 LOCATE HCP+I,16-(VCP+J+INT(I*FV*((FP*DD)/(FD*FP))):PRINT
CHR$(233)
1080 NEXT J
1090 NEXT I
1100 REM *****
1110 REM * AFFICHAGE DES RESULTATS *
1120 REM *****
1130 LOCATE 1,19:PRINT"- DECROISSANCE:";FP;"PAS DE";FV;"ET D'UN
E DUREE DE";FD;"CENTIEMES DE SECONDE CHACUNS."
1140 REM *****
1150 REM * CALCUL DE L'ORDRE BASIC CORRESPONDANT *
1160 REM *****
1170 LOCATE 1,21:INPUT "NUMERO DE CETTE ENVELOPPE S.V.P. ";NE
1180 IF NE<1 OR NE>16 THEN GOTO 1170
1190 LOCATE 1,21:PRINT "L'INSTRUCTION LUI CORRESPONDANT SERA DO
NC:"
1200 PRINT "ENV ";NE;",";AP;",";AV;",";AD;",";MP;",";MV;",";MD;
",";FP;",";FV;",";FD
1210 PRINT "ECOUTEZ LE SON OBTENU":
1220 ENV 1,AP,AV,AD,MP,MV,MD,FP,FV,FD
1230 REM *****
1240 REM * EXECUTION DE L'ORDRE OBTENU *
1250 REM *****
1260 FOR R=0 TO 2
1270 SOUND 1,119,0,0,1,0,0:SOUND 2,80,0,0,1,0,0
1280 FOR T=0 TO 500+DD:NEXT T
1290 NEXT R
1300 LOCATE 1,23:PRINT "POUR UNE AUTRE ENVELOPPE TAPER ENTER ."
1310 IF INKEY(18)=(0) THEN GOTO 60
1320 GOTO 1310
1330 REM *****
1340 REM * ROUTINE DE DEPLACEMENT DU CURSEUR *
1350 REM *****
1360 LOCATE HC,16-VC:PRINT CHR$(255)
1370 FOR T=1 TO 30:NEXT T
1380 LET HC=HC-(INKEY(1)=0)+(INKEY(8)=0)
1390 LET VC=VC+((INKEY(2)=0)-(INKEY(0)=0))*UP
1400 IF VC<1 THEN LET VC=1
1410 IF VC>15 THEN LET VC=15
1420 IF HC>50 THEN LET HC=50
1430 IF HC<1 THEN LET HC=1
1440 LOCATE HC,16-VC:PRINT CHR$(233)
1450 FOR T=1 TO 30: NEXT T
1460 IF INKEY(18)=0 THEN RETURN
1470 GOTO 1360

```

910.

La position du curseur en fin de maintien sera de nouveau mémorisée en utilisant HCP et VCP, UP prendra la valeur 0 puis la sous-routine 1360 sera engagée. La pente de la décroissance sera étudiée en fonction des valeurs prises par HC et VC par les tests des lignes 1010 et 1020 et le tracé débitera. Les résultats obtenus seront résumés par la ligne 1130.

Le programme pourra donc déterminer l'ordre basic correspondant à l'enveloppe tracée sur l'écran. Il demandera donc, en premier lieu, le numéro choisi pour celle-ci ; l'Amstrad pouvant en mémoriser 16. Dès lors l'instruction sera présentée au bas de l'écran, puis l'ensemble des variables réellement placé dans l'instruction ENV, de manière à vous présenter le résultat sonore obtenu. Le son sera alors émis trois fois — boucle des lignes 1260 à 1290 — et le programme en restera là jusqu'à ce que ENTER soit tapé pour un nouvel essai ou une autre enveloppe.

Pour la copie de frappe de ce programme et bien que nous ne l'ayons pas utilisé, il restera possible d'employer le basic abrégé.

Utilisation : après avoir tapé RUN, la grille s'affiche sur l'écran. Puis la phase : "DUREE TOTALE DE LA NOTE EN SECONDES ?" apparaît.

Un nombre supérieur à 0,5 devra être frappé, suivi de ENTER. Un résumé indiquera la durée de chaque division de l'écran ainsi que sa durée totale. Le curseur apparaîtra alors sur la grille et il sera possible de le déplacer à l'aide des flèches. Une fois la position désirée atteinte il suffira d'appuyer sur ENTER pour obtenir le tracé ainsi que les paramètres concernant l'attaque.

Il faudra répéter la même opération pour le maintien et la décroissance pour obtenir le gabarit complet de l'enveloppe.

Notons qu'il n'est pas indispensable d'utiliser l'ensemble de la surface de l'écran, pour pouvoir obtenir une enveloppe : si vous décidez finalement que la durée totale de votre son sera inférieure à celle présentée par l'écran, rien ne vous empêche de faire "finir" le son — par la commande de décroissance — en un point quelconque de l'écran. **Henri-Pierre PENEL**